Chapter 8

Differential Equations

STEWART'S CALCULUS: 5 ed. Ch. 10, 5ET ed. Ch. 9

This chapter focusses on the use of Maple commands to analyze differential equations. We use Maple's `dsolve` command to find explicit solutions to various first and second order differential equations. Maple is very good at finding explicit solutions for differential equations—when they can be found. However, solutions to many of the important differential equations cannot be found in closed form. In this case, the `numeric` option of `dsolve` can be used to find approximate numerical solutions to differential equations. In addition, the `DEplot` command can be used to plot direction fields, solution curves and phase portraits.

8.1 Explicit Solutions

STEWART'S CALCULUS: 5 ed. §§10.3 and 10.6, 5ET ed. §§9.3 and 9.6

To solve the first order, linear differential equation $y' + 5y = 2x$, first enter the equation as

```
> eq1:=diff(y(x),x)+5*y(x)=2*x;
```

$$eq1 := (\frac{d}{dx}\,\mathrm{y}(x)) + 5\,\mathrm{y}(x) = 2\,x$$

Notice that you must write `y(x)` and not `y` so that Maple knows `y` is a function of `x`. Then solve the equation using `dsolve`:

```
> dsolve(eq1,y(x));
```

$$\mathrm{y}(x) = \frac{2\,x}{5} - \frac{2}{25} + e^{(-5\,x)}\,\_C1$$

Notice the presence of the term `_C1`. This is Maple's notation for the arbitrary constant that occurs in the general solution of a first order differential equation. Also notice that the output is an equation (rather than an expression or a function or an assignment). So you need to use the **rhs** command to assign the right hand side to a label:

```
> ysol:=rhs(%);
```

$$ysol := \frac{2x}{5} - \frac{2}{25} + e^{(-5x)}\ \_C1$$

An initial condition such as $y(-1) = 2$ can be imposed in the following manner.

```
> dsolve({eq1,y(-1)=2},y(x));
```

$$\mathrm{y}(x) = \frac{2x}{5} - \frac{2}{25} + \frac{62}{25}\frac{e^{(-5x)}}{e^5}$$

Note that the equation and the initial condition are enclosed in curly braces. To see a plot of the solution, the following Maple commands can be used:

```
> ysol:=rhs(%):
> plot(ysol, x=-2..3, -1..4);
```



Notice that we have used the variables x and y here. However the independent variable is frequently taken as t to denote time.

The commands to solve a second (or higher) order differential equation are similar to the above. To find the general solution to the second order differential equation

$$\frac{d^2y}{dt^2} + 6\frac{dy}{dt} - 7y = 1$$

enter the commands

```
> eq2:=diff(y(t),t$2)+6*diff(y(t),t)-7*y(t)=1;
```

$$eq2 := \left(\frac{d^2}{dt^2}\,\mathrm{y}(t)\right) + 6\left(\frac{d}{dt}\,\mathrm{y}(t)\right) - 7\,\mathrm{y}(t) = 1$$

```
> dsolve(eq2,y(t));
```

$$\mathrm{y}(t) = e^{(-7t)}\ \_C2 + e^t\ \_C1 - \frac{1}{7}$$

Notice that there are two arbitrary constants in the solution, _C1 and _C2. To specify initial conditions such as $y(1) = 3$ and $y'(1) = 2$, enter

```
> dsolve({eq2, y(1)=3, D(y)(1)=2},y(t));
```

$$y(t) = \frac{1}{7}\, e^{(-7\,t)}\, e^{7} + 3\, e^{t}\, e^{(-1)} - \frac{1}{7}$$

Notice that the `diff` command cannot be used to specify an initial condition in `dsolve`. Rather, you must use the `D` operator and function notation.

8.2 Direction Fields

STEWART'S CALCULUS: 5 ed. §10.2, 5ET ed. §9.2

When you can't solve a differential equation. exactly, what do you do?

A direction field is a useful geometric device that aids in understanding the behavior of solutions to a first order differential equation, $\frac{dy}{dx} = F(x, y)$. At each point (x, y) on a solution $y = f(x)$, the slope of the solution curve is $m = f'(x) = F(x, y)$. The direction field is a way to display these slopes. At each point (x, y) draw a small line segment with slope $m = F(x, y)$. Then any solution curve will be everywhere tangent to these line segments.

Maple can plot direction fields using the `DEplot` command from the `DEtools` package. To plot the direction field for the differential equation $y' = x\sin(y)$, first load the `DEtools` package and define the equation:

```
> with(DEtools):
> deq:=diff(y(x),x)=x*sin(y(x));
```

$$deq := \frac{d}{dx}\,\mathrm{y}(x) = x\sin(\mathrm{y}(x))$$

Then use the `DEplot` command:

```
> DEplot(deq, y(x), x=-2..2, y=-1..1);
```



To graph one or more solutions to the differential equation together with the direction field, you need to add the initial conditions. Suppose you want three solution curves with initial conditions $f(0) = 0.5$, $f(1) = -0.5$ and $f(-1) = 0.25$. Then you execute

```
> inits:=[[0,.5], [1,-.5], [-1,.25]];
```

$$inits := [[0, 0.5], [1, -0.5], [-1, 0.25]]$$

```
> DEplot(deq, y(x), x=-2..2, y=-1..1, inits);
```



Notice how the solution curves are everywhere tangent to the direction field. In fact, this tangency is the basic idea behind Euler's method, which is a numerical algorithm for calculating approximate solutions to differential equations.

Finally, to plot the curves without the direction field, add the option `arrows=none`:

```
> DEplot(deq, y(x), x=-2..2, y=-1..1, inits, arrows=none);
```



8.3 Numerical Solutions

STEWART'S CALCULUS: 5 ed. §10.2, 5ET ed. §9.2

The material in this section is optional. However, since many nonlinear differential equations are virtually impossible to solve (in closed form), students should find the information in this section very useful.

Maple can find numerical approximations to solutions to differential equations. The algorithm that Maple uses is a Fehlberg fourth-fifth order Runge-Kutta method, which is more sophisticated than Euler's method. For example, suppose we want to solve $y' + \sin(y^2) = 1$, with $y(0) = 1$. If we proceed as in the previous section, we get the following.

```
> deq:=diff(y(t),t)+sin(y(t)^2)=1;
```

$$deq := \left(\frac{d}{dt}\,y(t)\right) + \sin(y(t)^2) = 1$$

```
> dsolve({deq,y(0)=1},y(t));
```

$$y(t) = \mathrm{RootOf}\left(t + \int_0^{\_Z} \frac{1}{\sin(\_a^2) - 1}\,d\_a - \int_0^1 \frac{1}{\sin(\_a^2) - 1}\,d\_a\right)$$

Maple is telling us that it cannot evaluate some integral in closed form.

To obtain an approximate solution using `dsolve`, add the `numeric` option. (There must be an initial condition.) Maple's output is then a procedure, which can be used in a manner similar to a Maple function. (For details on procedures, see Chapter 10.)

```
> f:=dsolve({deq,y(0)=1},y(t),numeric);
```

$$f := \textbf{proc}(x\_rkf45)\ \ldots\ \textbf{end proc}$$

We have assigned the label `f` to this procedure so that we can use it to find approximate values for the solution at various values of t. We evaluate it just like a function. Its values at $t = 0$ and $t = 0.5$ are:

```
> f(0); f(0.5);
```

$$[t = 0., \ y(t) = 1.]$$

$$[t = 0.5, \ y(t) = 1.06203222331694192]$$

To plot the solution we use Maple's `odeplot` command in the `plots` package:

```
> with(plots):
> odeplot(f,[t,y(t)],0..4);
```



NOTE: The variables in the list $[t, y(t)]$ must be the same symbols for the independent and dependent variables that were used when the procedure `f` was defined. Alternatively, the plot can be obtained directly by using Maple's `DEplot` command in the `DEtools` package:

```
> with(DEtools):
> DEplot(deq, y(t), t=0..4, [[0,1]], arrows=none);
```